DİKDÖRTGEN MATRİSİN TERSİNİN BULUNMASI PROBLEMİNE BİR YAKLAŞIM

Yrd. Doç.Dr. Hasan DURUCASU
Anadolu Üniversitesi
İktisadi ve İdari Bilimler Fakültesi
İşletme Bölümü

ABSTRACT

Matrices are mathematical elements which are frequently applied in various sciences. Linear equation systems are used in the field of business administration under the topics of break even analysis, selection of production method, Marchow's chains, etc,. One of the solving methods of linear equation systems is the use of matrix approach. The solution of linear equations system which is expressed as $\mathbf{AX} = \mathbf{B}$ is found as $\mathbf{X} = \mathbf{A}^{-1}\mathbf{B}$. The term $\mathbf{A}^{-1}$ which take place in the last expression is the inverse of coefficient matrix $\mathbf{A}$ related to linear equations system. If $\mathbf{A}$ is a square matrix (number of rows equals number of columns), it is known that the inverse can be found in the condition that $|\mathbf{A}| \neq 0$. In the case that the number of rows and columns are different, this matrix is now called as a rectangular matrix and the operation of finding the inverse matrix becomes more difficult.

In high order matrices to eliminate the boredom of repetead simple hand-made mathematical calculations and especially to simplify the operations in finding inverse of rectangular matrices, the use of computer has been suggested. In this study, for this purpose, each given formulation of rectangular matrix inversion is followed by related computer programs.

ÖZET

Matrisler değişik bilim dallarında yaygın uygulama alanı bulan matematik elemanlardır. İşletme alanında başabaş analizi, üretim yönteminin seçimi, Markov zincirleri, v.b...gibi konularda doğrusal denklem sistemleri kullanılmaktadır. Doğrusal denklem sistemlerinin çözüm yöntemlerinden birini de matris kullanımı yaklaşımı oluşturmaktadır. $\mathbf{AX} = \mathbf{B}$ biçiminde matrisler yardımıyla ifade edilen bir doğrusal denklem sisteminin çözümü $\mathbf{X} = \mathbf{A}^{-1}\mathbf{B}$ olarak bulunabilmektedir. Son ifade de yer alan $\mathbf{A}^{-1}$ terimi doğrusal denklem sistemine ilişkin katsayılar matrisi olan $\mathbf{A}$ ' nın tersidir. $\mathbf{A}$ kare matris (satır sayısı sütun sayısına eşit) olduğunda, tersinin $|\mathbf{A}| \neq 0$ koşuluyla bulunabildiği bilinmektedir. Bir matrisin satır ve sütun sayıları farklı olduğunda bu matris dikdörtgen matris adını almakta ve bu matrisin tersinin bulunmasına ilişkin işlemler daha da zorlaşmaktadır. Yüksek mertebeli matrislerde, elle yapılan tekrarlı basit aritmetik işlemlerin sıkıcılığının ortadan kaldırılması ve formülasyonunun güçlüğü nedeniyle özellikle dikdörtgen matrislerin terslerinin kolayca bulunabilmasi için bilgisayardan yararlanılması düşünülmüştür. Bu nedenle bu çalışmada verilen her dikdörtgen ters matris alma formülasyonunu ilgili bilgisayar programları izlemiştir.

GİRİŞ

m satırlı ve n sütunlu $\mathbf{A}$ matrisinin tersinin bulunması için gerekli işlemler m ve n değerlerine göre farklılık gösterir. $m = n$ olması durumunda $\mathbf{A}$ matrisi kare matris adını alır ve $\mathbf{I}$ (esas köşegen elemanları 1, diğerleri 0 olan) birim matrisi göstermek üzere

$$\mathbf{A}\mathbf{A}^{-1} = \mathbf{I} \qquad (1)$$

$$\mathbf{A}^{-1}\mathbf{A} = \mathbf{I} \qquad (2)$$

bağıntılarını sağlayan $\mathbf{A}^{-1}$ matrisi bulunabilir. (1)'de yer alan $\mathbf{A}^{-1}$ sağ ters matris adını alırken, (2)'deki $\mathbf{A}^{-1}$ matrisi sol ters matristir. Ters matris alma işlemleri yardımıyla, m denklemli ve n bilinmeyenli ($m = n$) doğrusal denklem sisteminin çözümü gerçekleştirildiğinde, Kofaktörler ya da Gauss

Eleme metodlarını temel alan alışılmış algoritmalar **A** kare matrisinin sol ters matrisini hesaplamaktadır.

Satır sayısı m'nin sütun sayısı n'den farklı olduğu dikdörtgen matrislerin tersleri söz konusu olduğunda, sağ ya da sol ters matris kavramları daha da önem kazanmakta ve **A** matrisi üzerinde $m < n$ veya $m > n$ olmasına bağlı olarak birbirine çok benzeyen farklı işlem adımları uygulanmaktadır.

I. Durum: Satır sayısının sütun sayısından küçük olması durumu

m satırlı ve n sütunlu $(m < n)$ **A** dikdörtgen matrisi verildiğinde, bu matris ${}_k\mathbf{A}$ olarak gösterilsin. Bu durumda $({}_k\mathbf{A})^{-1}_{mxn}$ ifadesi ise ${}_k\mathbf{A} = (\mathbf{a}_{ij})_{mxn}$ matrisinin ters matrisine ilişkin bir gösterim olacaktır. Öte yandan m satırlı ve n sütunlu $(m < n)$ ${}_k\mathbf{A}$ matrisi açık olarak

$$
{}_k\mathbf{A} = \begin{bmatrix} \mathbf{a}_{1,1} & \mathbf{a}_{1,2} & \cdots & \mathbf{a}_{1,m} & \cdots & \mathbf{a}_{1,n} \\ \mathbf{a}_{2,1} & \mathbf{a}_{2,2} & \cdots & \mathbf{a}_{2,m} & \cdots & \mathbf{a}_{2,n} \\ \cdots & \cdots & \cdots & \cdots & \cdots & \cdots \\ \mathbf{a}_{m,1} & \mathbf{a}_{m,2} & \cdots & \mathbf{a}_{m,m} & & \mathbf{a}_{m,n} \end{bmatrix}
$$
biçiminde yazılır.

${}_k\mathbf{A} = (\mathbf{a}_{ij})_{mxn}$ $(m < n)$ matrisinin $a_{1,1}$ elemanından başlanarak oluşturulan mxm mertebeli alt kare matris ${}_{k-1}\mathbf{A}$ ile gösterilsin. Söz konusu bu matrisin açık ifadesi de

$$
{}_{k-1}\mathbf{A} = \begin{bmatrix} \mathbf{a}_{1,1} & \mathbf{a}_{1,2} & \cdots & \mathbf{a}_{1,m} \\ \mathbf{a}_{2,1} & \mathbf{a}_{2,2} & \cdots & \mathbf{a}_{2,m} \\ \cdots & \cdots & \cdots & \cdots \\ \mathbf{a}_{m,1} & \mathbf{a}_{m,2} & \cdots & \mathbf{a}_{m,m} \end{bmatrix}
$$

biçiminde olacaktır.

${}_k\mathbf{A}$ matrisinin, ${}_{k-1}\mathbf{A}$ matrisi dışındaki bölümü olan $mx(n-m)$ mertebeli matris ${}_k\mathbf{a}$ olarak isimlendirilsin. Bu durumda, ${}_k\mathbf{a}$ matrisi de yukarıdaki açık matris yazılımlarına benzer biçimde

$$ {}_{k}a = \begin{bmatrix} a_{1,m+1} & a_{1,m+2} & \cdots & a_{1,n} \\ a_{2,m+1} & a_{2,m+1} & \cdots & a_{2,n} \\ \cdots & \cdots & \cdots & \cdots \\ a_{m,m+1} & a_{m,m+2} & \cdots & a_{m,n} \end{bmatrix} $$ olarak yazılabilecektir.

Böylece, mxn mertebeli ${}_{k}A$ dikdörtgen matrisinin, mxm mertebeli ${}_{k-1}A$ kare matrisi ve $mx(n-m)$ mertebeli ${}_{k}a$ dikdörtgen matrisi cinsinden , bölümlendirilmiş matris yazılımı kullanılarak ifadesi,

${}_{k}A = \left[{}_{k-1}A \quad {}_{k}a\right]$ biçiminde özetlenebilecektir.

${}_{k-1}A$, mxm mertebeli olduğundan, ${}_{k-1}A$ 'nın tersi olan $({}_{k-1}A)^{-1}$ bilinen kare matris tersi hesaplama yollarıyla bulunur. Kolayca izlenebileceği gibi $({}_{k-1}A)^{-1}$ matrisi de mxm mertebeli bir kare matristir.

$({}_{k-1}A)^{-1}$ bulunduktan sonra evriği olan $\left\{({}_{k-1}A)^{-1}\right\}^{T}$ mxm mertebeli kare matrisi sütunlarla satırların yerlerinin değiştirilmesiyle elde edilir.

Öte yandan $({}_{k-1}A)^{-1}({}_{k}a)$ matris çarpımının normunun, iç çarpımdan faydalanılarak $\left\| {}_{k-1}A)^{-1}({}_{k}a)\right\| < ({}_{k-1}A)^{-1}({}_{k}a),({}_{k-1}A)^{-1}({}_{k}a) >$ olarak yazılabileceği bilinmektedir. Yukarıdaki son ifadede yer alan $({}_{k-1}A)^{-1}({}_{k}a)$ matris çarpımı da, elemanları cinsinden

$$ ({}_{k-1}A)^{-1}({}_{k}a) = (a_{i,j}) \qquad (i = 1,\ldots,m;\ j = 1,\ldots,n-m) $$

olarak ifade edilir. Son eşitlik göz önünde tutularak iç çarpım ifadesi

$$ <({}_{k-1}A)^{-1}({}_{k}a),({}_{k-1}A)^{-1}({}_{k}a) > = < (a_{i,j})(a_{i,j}) > $$

biçimine girer. Bu son ifadeden hareket edilerek işlemler ilerletildiğinde

$$<({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a}),({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a})>=<(\mathbf{a}_{i,j})(\mathbf{a}_{i,j})>=\sum_{i,j=1}^{m,n-m}\mathbf{a}_{i,j}\mathbf{a}_{i,j}=\sum_{i,j=1}^{m,n-m}\mathbf{a}_{i,j}^{2} \quad (3)$$

elde edilir. (3) 'ün sağ yanı açık olarak

$$\begin{aligned}<({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a}),({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a})>=\;&\mathbf{a}_{1,1}^{2}+\mathbf{a}_{1,2}^{2}+\ldots+\mathbf{a}_{1,n-m}^{2}+\\&+\mathbf{a}_{2,1}^{2}+\mathbf{a}_{2,2}^{2}+\ldots+\mathbf{a}_{2,n-m}^{2}+\\&\quad+\qquad+\\&+\mathbf{a}_{m,1}^{2}+\mathbf{a}_{m,2}^{2}+\ldots+\mathbf{a}_{m,n-m}^{2}\end{aligned} \quad (4)$$

yazıldığında, sağ yanın reel bir sayı olduğu kolayca görülür.

Algoritmanın bir sonraki işlem adımını, $\left\{({}_{k-1}\mathbf{A})^{-1}\right\}^{T}({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a})$ üçlü matris çarpımının oluşturulup (4)'te hesaplanan reel sayının bir fazlasına bölünmesi yoluyla (5) 'te verilen bir ${}_{k}\mathbf{P}$ matrisinin elde edilmesi oluşturur.

$${}_{k}\mathbf{P}=\frac{\left\{({}_{k-1}\mathbf{A})^{-1}\right\}^{T}({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a})}{<({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a}),({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{a})>+1} \quad (5)$$

Bundan sonra, (5) 'deki ${}_{k}\mathbf{P}$ matrisinin evriği $({}_{k}\mathbf{P})^{T}$ bilinen işlemlerle kolayca bulunur.

Daha sonra $({}_{k}\mathbf{a})({}_{k}\mathbf{P})^{T}$ matris çarpımı oluşturulur ve birim matristen $({}_{k}\mathbf{a})({}_{k}\mathbf{P})^{T}$ çarpımın çıkarılmasıyla (6) 'da verilen ${}_{k}\mathbf{B}$ matrisi elde edilir.

$${}_{k}\mathbf{B}=\mathbf{I}-({}_{k}\mathbf{a})({}_{k}\mathbf{P})^{T} \quad (6)$$

(6) 'daki ${}_{k}\mathbf{B}$ matrisinin $({}_{k-1}\mathbf{A})^{-1}$ matrisi ile soldan çarpımı ile (7) 'deki ${}_{k}\mathbf{Y}$ matrisi oluşur.

$$ {}_{k}\mathbf{Y} = ({}_{k-1}\mathbf{A})^{-1}({}_{k}\mathbf{B}) \tag{7} $$

Nihayet $({}_{k}\mathbf{P})^{T}$ ve ${}_{k}\mathbf{Y}$ matrislerinin (8)'deki biçimde düzenlenmesiyle bulunması amaçlanan $({}_{k}\mathbf{A})^{-1}$ matrisi elde edilir.

Böylelikle ${}_{k}\mathbf{A}$ matrisinin tersi

$$ ({}_{k}\mathbf{A})^{-1} = \begin{bmatrix} {}_{k}\mathbf{Y} \\ {}_{k}\mathbf{P}^{T} \end{bmatrix} \tag{8} $$

olarak elde edilir.

(8) uyarınca bulunan $({}_{k}\mathbf{A})^{-1}$, ${}_{k}\mathbf{A}$ dikdörtgen matrisinin sağ tersidir. Diğer bir ifadeyle,

$$ ({}_{k}\mathbf{A})({}_{k}\mathbf{A})^{-1} = \mathbf{I} $$

bağıntısını sağlayan $({}_{k}\mathbf{A})^{-1}$ matrisi bulunmuştur.

Algoritmayı temel alan program listesi izleyen kesimde yer almaktadır.

```
10 REM M<N HALİNDE TERS MATRIS (SAĞ TERS) BULAN PROGRAM
20 CLS
30 READ M,N
40 DIM A(M,N),B(M,M),C(M,N-M),D(M,M),E(M,M),F(M,N-M),G(M,N-
M),H(M,N-M),P(N-M,M),Q(M,M),R(M,M),S(M,M),T(M,M),Z(N,M),
DETERATB(M,M)
50 REM A MATRİSİNİN OKUTULUP YAZDIRILMASI
60 PRINT: PRINT: PRINT "VERİLEN MATRIS": PRINT " ----------------":
PRINT
70 FOR I =1 TO M
80 FOR J =1 TO N:READ A(I, J): PRINT USING "###.######";A(I,J);:
NEXT J
90 PRINT
100 NEXT I
```

```
110 REM A MATRISI İÇİNDE YER ALAN EN BÜYÜK KARE MATRISIN B'DE OLUŞTURULMASI
120 FOR I =1 TO M
130 FOR J =1 TO M: B(I,J) = A(I,J): NEXT J
140 NEXT I
150 REM A 'NIN EN BÜYÜK KARE MATRISI B'NIN A'DAN ÇIKARILMASI
160 REM SONRASI KALAN MATRİS BÖLÜMÜNÜN C'YE AKTARILMASI
170 FOR I =1 TO M
180 FOR J =M+1 TO N: C(I,J-M)=A(I,J): NEXT J
190 NEXT I
200 REM B'NIN TERSİNİN YİNE B'DE OLUŞTURULUP D'YE AKTARILMASI
210 FOR I =1 TO M
220 FOR J =1 TO M
230 DETERB (I,J)=B(I,J)
240 NEXT J
250 NEXT I
260 FOR K=1 TO M-1
270 FOR I=K+1 TO M
280 X =DETERB(I,K)/DETERB(K,K)
290 FOR J=K+1 TO M
300 DETERB(I,J)=DETERB(I, J)-X*DETERB(K,J)
310 NEXT J
320 NEXT I
330 NEXT K
340 DET=1!
350 FOR I=1 TO M
360 DET =DET*DETERB(I,I)
370 NEXT I
380 IF DET=0 THEN PRINT "EN BÜYÜK KARE MATRİSİN DETERMİNANTI SIFIRA EŞİT TERS MATRİS BULUNAMAZ": GOTO 1550
390 FOR K=1 TO M
400 FOR I =1 TO M
410 FOR J =1 TO M
420 IF I=K OR J=K THEN 440
430 B(I,J)=B(I,J)-B(I,K)*B(K,J)/B(K,K)
440 NEXT J
450 NEXT I
460 B(K,K) =-1!/B(K,K)
470 FOR I= 1 TO M
480 IF I=K THEN 500
490 B(I,K)=B(I,K)*B(K,K)
500 NEXT I
```

```
510 FOR J =1 TO M
520 IF J=K THEN 540
530 B(K,J)=B(K,J)*B(K,K)
540 NEXT J
550 NEXT K
560 FOR I=1 TO M
570 FOR J=1 TO M
580 B(I,J)=-B(I,J)
590 NEXT J
600 NEXT I
610 FOR I=1 TO M
620 FOR J=1 TO M
630 D(I,J)=B(I,J)
640 NEXT J
650 NEXT I
660 REM D'NIN TRANSPOZESİ E'NİN BELLEKTE OLUŞTURULMASI
670 FOR I=1 TO M
680 FOR J=1 TO M: E(I,J)=B(J,I): NEXT J
690 NEXT I
700 REM D VE C MATRİSLERİNİN ÇARPIM MATRİSİ F'NİN BELLEKTE
OLUŞTURULMASI
710 FOR I=1 TO M
720 FOR J=1 TO N-M
730 FOR K=1 TO M: F(I,J)=F(I,J)+D(I,K)*C(K,J): NEXT K
740 NEXT J
750 NEXT I
760 REM E VE F MATRİSLERİNİN ÇARPIM MATRİSİ G'NİN
OLUŞTURULMASI
770 FOR I=1 TO M
780 FOR J=1 TO N-M
790 FOR K=1 TO M: G(I,J)=G(I,J)+E(I,K)*F(K,J): NEXT K
800 NEXT J
810 NEXT I
820 REM G MATRİSİNİN NORMUNUN HESAPLANMASI
830 TOP=0
840 FOR I=1 TO M
850 FOR J=1 TO N-M
860 TOP =TOP+G(I,J)^ 2
870 NEXT J
880 NEXT I
890 REM G MATRİSİNİN NORMUNA 1 EKLENMESİ
900 TOP=TOP+1
```

```
910 REM G MATRİSİNİN NORM + 1'E BÖLÜNEREK H MATRİSİNİN
OLUŞTURULMASI
920 FOR I=1 TO M
930 FOR J=1 TO N-M
940 H(I,J)=G(I,J)/TOP
950 NEXT J
960 NEXT I
970 REM H MATRİSİNİN TRANSPOZESİ P MATRİSİNİN
OLUŞTURULMASI
980 FOR I=1 TO N-M
990 FOR J=1 TO M
1000 P(I,J)=H(J,I)
1010 NEXT J
1020 NEXT I
1030 REM C VE P'NİN ÇARPIM MATRİSİ Q'NUN OLUŞTURULMASI
1040 FOR I=1 TO M
1050 FOR J=1 TO M
1060 FOR K=1 TO N-M: Q(I,J)= Q(I,J)+C(I,K)*P(K,J): NEXT K
1070 NEXT J
1080 NEXT I
1090 REM M X M'LİK BİRİM MATRİS OLUŞTURMA
1100 FOR I=1 TO M
1110 FOR J=1 TO M
1120 IF I=J THEN R(I,J)=1 ELSE R(I,J)=0
1130 NEXT J
1140 NEXT I
1150 REM R - Q'NUN S DE OLUŞTURULMASI
1160 FOR I=1 TO M
1170 FOR J=1 TO M
1180 S(I,J)=R(I,J) - Q(I,J)
1190 NEXT J
1200 NEXT I
1210 REM A'NIN EN BÜYÜK KARE MATRİSİ OLAN B'NİN TERSİYLE
S'NİN ÇARPIM MATRİSİ OLAN T'NİN OLUŞTURULMASI
1220 FOR I=1 TO M
1230 FOR J=1 TO M
1240 FOR K=1 TO N-M: T(I,J)=T(I,J)+ D(I,K)*S(K,J): NEXT K
1250 NEXT J
1260 NEXT I
1270 REM HESAPLANAN T VE P'NİN YERLEŞTİRİLMESİYLE Z TERS
MATRİSİNİN OLUŞTURULMASI
1280 REM 1) T MATRİSİNİN Z'YE AKTARILMASI
1290 FOR I=1 TO M
```

```
1300 FOR J=1 TO M
1310 Z(I,J)=T(I,J)
1320 NEXT J
1330 NEXT I
1340 REM 2) Z'DEKİ T MATRİSİNİN ALTINA P MATRİSİNİN
YERLEŞTİRİLMESİ
1350 FOR I=M+1 TO N
1360 FOR J=1 TO M
1370 Z(I,J)=P(I-M,J)
1380 NEXT J
1390 NEXT I
1400 PRINT: PRINT: PRINT "TERS MATRİS": PRINT "............................"
1410 REM (mxn)'LIK A'NIN TERS MATRİSİ (nxm)'LIK Z'NIN EKRANDA
GÖRÜNTÜLENMESİ
1420 FOR I=1 TO N
1430 FOR J=1 TO M
1440 PRINT USING "###.######"; Z(I,J);
1450 NEXT J
1460 PRINT
1470 NEXT I
1480 DATA 3,5
1490 DATA  1, 0, -1, 0,-2
1500 DATA -2, 1,  3, 1, 2
1510 DATA  1, 1,  1, 0, 1
1550 END
```

Programın çalıştırılmasına ilişkin örnek çıktı aşağıda verilmiştir.

Verilen Matris

1.000000	0.000000	-1.000000	0.000000	-2.000000
-2.000000	1.000000	3.000000	1.000000	2.000000
1.000000	1.000000	1.000000	0.000000	1.000000

Ters Matris

-1.969197	-0.987073	-0.000246
4.911230	1.965074	0.023769
-2.942033	-0.977201	-0.15523
0.003630	0.001455	-0.000970
-0.013582	-0.005336	0.003630

II.Durum: Satır sayısının sütun sayısından büyük olması durumu

Tersi alınması istenen m satırlı ve n sütunlu özgün matris $m > n$ olarak verildiğinde, ters matrisi bulmaya yönelik algoritma I.Durumda yapılan işlemlerin benzerlerini içerecektir. İki durum arasındaki temel farklılık matris çarpımlarında kendini gösterecek, matris çarpım işlemleri I.Durum'da olduğu gibi soldan sağa doğru değil de, bu kez sağdan sola doğru gerçekleştirilecektir.

Tersi alınacak m satırlı ve n sütunlu (m>n) matris bu kez $\mathbf{A}_k$ olarak ifade edilip, açık olarak

$$\mathbf{A}_k = \begin{bmatrix} a_{1,1} & a_{1,2} & \dots & a_{1,n} \\ a_{2,1} & a_{2,2} & \dots & a_{2,n} \\ \dots & \dots & \dots & \dots \\ a_{n,1} & a_{n,2} & \dots & a_{n,n} \\ \dots & \dots & \dots & \dots \\ a_{m,1} & a_{m,2} & \dots & a_{m,n} \end{bmatrix}$$

biçiminde yazılacaktır.

$\mathbf{A}_k = (\mathbf{a}_{i,j})_{mxn}$ (m>n) matrisinin $a_{1,1}$ elemanından başlanarak oluşturulan nxn mertebeli alt kare matris bu durumda $\mathbf{A}_{k-1}$ ile gösterilsin. Bu kare matris açık olarak

$$\mathbf{A}_{k-1} = \begin{bmatrix} a_{1,1} & a_{1,2} & \dots & a_{1,n} \\ a_{2,1} & a_{2,2} & \dots & a_{2,n} \\ \dots & \dots & \dots & \dots \\ a_{n,1} & a_{n,2} & \dots & a_{n,n} \end{bmatrix}$$

biçiminde yazılır.

$\mathbf{A}_k$ matrisinden, $\mathbf{A}_{k-1}$ alt matrisi dışarıya alındıktan sonraki bölümü $\mathbf{a}_k$ ile ifade edildiğinde,

$$\mathbf{a}_k = \begin{bmatrix} \mathbf{a}_{n+1,1} & \mathbf{a}_{n+1,2} & \cdots & \mathbf{a}_{n+1,n} \\ \mathbf{a}_{n+2,1} & \mathbf{a}_{n+2,2} & \cdots & \mathbf{a}_{n+2,n} \\ \cdots & \cdots & \cdots & \cdots \\ \mathbf{a}_{m,1} & \mathbf{a}_{m,2} & \cdots & \mathbf{a}_{m,n} \end{bmatrix}$$

olarak yazılır.

Böylelikle mxn mertebeli $\mathbf{A}_k$ dikdörtgen matrisi, nxn mertebeli $\mathbf{A}_{k-1}$ kare matrisi ve $(m-n)xn$ mertebeli $\mathbf{a}_k$ dikdörtgen matrisi cinsinden , bölümlendirilmiş matris yazılımıyla,

$$\mathbf{A}_k = \begin{bmatrix} \mathbf{A}_{k-1} \\ \mathbf{a}_k \end{bmatrix}$$

biçiminde ifade edilir.

Bir sonraki işlem adımında $\mathbf{A}_{k-1}$ kare matrisinin tersi bulunur ve evriği olan $\{(\mathbf{A}_{k-1})^{-1}\}^T$ alınır. Kolayca izlenebileceği gibi bu son matris nxn mertebelidir.

Sonraki işlem olarak, $\mathbf{a}_k(\mathbf{A}_{k-1})^{-1}$ matris çarpımının normunun iç çarpımdan faydalanılarak hesaplanması amacıyla,

$$\left\|\mathbf{a}_k(\mathbf{A}_{k-1})^{-1}\right\|^2 = \left(\sqrt{<\mathbf{a}_k(\mathbf{A}_{k-1})^{-1}, \mathbf{a}_k(\mathbf{A}_{k-1})^{-1}>}\right)^2 =< \mathbf{a}_k(\mathbf{A}_{k-1})^{-1}, \mathbf{a}_k(\mathbf{A}_{k-1}$$

formülünden hareket edilerek ve yukarıdaki ifadenin en sağında yer alan yazılımdaki $\mathbf{a}_k(\mathbf{A}_{k-1})^{-1}$ matris çarpımının elemanları cinsinden

$$(\mathbf{a}_k)(\mathbf{A}_{k-1})^{-1} = (\mathbf{a}_{i,j}) \qquad (i = 1,\ldots,m-n;\ j = 1,\ldots,n)$$

olduğu göz önünde bulundurularak,

$$\left\|\mathbf{a}_k(\mathbf{A}_{k-1})^{-1}\right\|^2 =< \mathbf{a}_k(\mathbf{A}_{k-1})^{-1}, \mathbf{a}_k(\mathbf{A}_{k-1})^{-1} >= \sum_{i,j=1}^{m-n,n} \mathbf{a}_{ij}\mathbf{a}_{ij} = \sum_{i,j=1}^{m-n,n} \mathbf{a}^2_{ij} \quad (9)$$

olarak yazılabileceği bilinmektedir. (9) 'un sağ yanı bir reel sayıdır.

İzleyen işlem adımında, $(\mathbf{a}_k)(\mathbf{A}_{k-1})^{-1}\left\{(\mathbf{A}_{k-1})^{-1}\right\}^T$ üçlü matris çarpımının oluşturulup (7)'de elde edilen reel sayının bir fazlasına bölünmesi yoluyla (10) 'da verilen $\mathbf{P}_k$ matrisi elde edilir.

$$\mathbf{P}_k = \frac{(\mathbf{a}_k)(\mathbf{A}_{k-1})^{-1}\left\{(\mathbf{A}_{k-1})^{-1}\right\}^T}{\left\|\mathbf{a}_k)(\mathbf{A}_{k-1})^{-1}\right\|^2 + 1} \tag{10}$$

Bundan sonra, (10) 'daki $\mathbf{P}_k$ matrisinin evriği $(\mathbf{P}_k)^T$ bulunur. Kolayca algılanabileceği gibi $(\mathbf{P}_k)^T$ $(m-n)xn$ mertebeli bir matristir.

Sonraki işlem adımında, $(\mathbf{P}_k)^T(\mathbf{a}_k)$ çarpımından oluşan matrisin, nxn mertebeli birim matristen çıkarılmasıyla oluşan matrisin, sağdan $(\mathbf{A}_{k-1})^{-1}$ matrisiyle çarpımı olan

$$\mathbf{Y}_k = \left(\mathbf{I} - (\mathbf{P}_k)^T(\mathbf{a}_k)\right)(\mathbf{A}_{k-1})^{-1}$$

matrisi bulunur.

Nihayet $(\mathbf{P}_k)^T$ ve $\mathbf{Y}_k$ matrislerinin $\left[\mathbf{Y}_k \quad \mathbf{P}_k^T\right]$ biçiminde düzenlenmesiyle aranan ters matris

$$(\mathbf{A}_k)^{-1} = \left[\mathbf{Y}_k \quad \mathbf{P}_k^T\right]$$

olarak bulunur.

Yukarıda açıklanmaya çalışılan biçimde bulunan $(\mathbf{A}_k)^{-1}$ matrisi mxn mertebeli $\mathbf{A}_k$ dikdörtgen matrisinin sol tersidir ve

$$(\mathbf{A}_k)^{-1}\mathbf{A}_k = \mathbf{I}$$

eşitliğini sağlar.

İlgili program listesi ve örnek çıktısı izleyen kesimde yer almaktadır.

```
10 REM M> N HALİNDE TERS MATRIS (SOL TERS) BULAN PROGRAM
20 CLS
30 READ M,N
40 DIM A(M,N),B(N,N),C(M-N,N),D(N,N),E(N,N),F(M-N,N),G(M-N,N),H(M-
N,N),P(N,M-N),Q(N,N),R(N,N),S(N,N),T(N,N),Z(N,M), DETERDB(N,N)
50 REM A MATRİSİNİN OKUTULUP YAZDIRILMASI
60 PRINT: PRINT: PRINT "VERİLEN MATRIS": PRINT " -------------":
PRINT
70 FOR I =1 TO M
80 FOR J =1 TO N:READ A(I, J): PRINT USING "###.######";A(I,J);:
NEXT J
90 PRINT
100 NEXT I
110 REM A MATRISI İÇİNDE YER ALAN EN BÜYÜK KARE MATRISIN
B'DE OLUŞTURULMASI
120 FOR I =1 TO N
130 FOR J =1 TO N: B(I,J) = A(I,J): NEXT J
140 NEXT I
150 REM A MATRİSİNİN EN BÜYÜK KARE MATRISI B'NİN A'DAN
ÇIKARILMASI SONRASI
160 REM KALAN MATRİS BÖLÜMÜNÜN C'YE AKTARILMASI
170 FOR I =N+1 TO M
180 FOR J =1 TO N: C(I-N,J)=A(I,J): NEXT J
190 NEXT I
200 REM B'NIN TERSİNİN YİNE B'DE OLUŞTURULUP D'YE
AKTARILMASI
210 FOR I =1 TO N
220 FOR J =1 TO N
230 DETERB (I,J)=B(I,J)
240 NEXT J
250 NEXT I
260 FOR K=1 TO N-1
270 FOR I=K+1 TO N
280 X =DETERB(I,K)/DETERB(K,K)
290 FOR J=K+1 TO N
300 DETERB(I,J)=DETERB(I, J)-X*DETERB(K,J)
310 NEXT J
320 NEXT I
330 NEXT K
340 DET=1!
350 FOR I=1 TO N
360 DET =DET*DETERB(I,I)
```

```
370 NEXT I
380 IF DET=0 THEN PRINT "EN BÜYÜK KARE MATRİSİN
DETERMİNANTI SIFIRA EŞİT TERS MATRİS BULUNAMAZ": GOTO 1550
390 FOR K=1 TO N
400 FOR I =1 TO N
410 FOR J =1 TO N
420 IF I=K OR J=K THEN 440
430 B(I,J)=B(I,J)-B(I,K)*B(K,J)/B(K,K)
440 NEXT J
450 NEXT I
460 B(K,K) =-1!/B(K,K)
470 FOR I= 1 TO N
480 IF I=K THEN 500
490 B(I,K)=B(I,K)*B(K,K)
500 NEXT I
510 FOR J =1 TO N
520 IF J=K THEN 540
530 B(K,J)=B(K,J)*B(K,K)
540 NEXT J
550 NEXT K
560 FOR I=1 TO N
570 FOR J=1 TO N
580 B(I,J)=-B(I,J)
590 NEXT J
600 NEXT I
610 FOR I=1 TO N
620 FOR J=1 TO N
630 D(I,J)=B(I,J)
640 NEXT J
650 NEXT I
660 REM D'NIN TRANSPOZESİ E'NİN BELLEKTE OLUŞTURULMASI
670 FOR I=1 TO N
680 FOR J=1 TO N: E(I,J)=D(J,I): NEXT J
690 NEXT I
700 REM C VE D MATRİSLERİNİN ÇARPIM MATRİSİ F'NNİN BELLEKTE
OLUŞTURULMASI
710 FOR I=1 TO M-N
720 FOR J=1 TO N
730 FOR K=1 TO N: F(I,J)=F(I,J)+C(I,K)*D(K,J): NEXT K
740 NEXT J
750 NEXT I
760 REM F VE E MATRİSLERİNİN ÇARPIM MATRİSİ G'NİN
OLUŞTURULMASI
```

```
770 FOR I=1 TO M-N
780 FOR J=1 TO N
790 FOR K=1 TO N: G(I,J)=G(I,J)+F(I,K)*E(K,J): NEXT K
800 NEXT J
810 NEXT I
820 REM G MATRİSİNİN NORMUNUN HESAPLANMASI
830 TOP=0
840 FOR I=1 TO M-N
850 FOR J=1 TO N
860 TOP =TOP+G(I,J)^ 2
870 NEXT J
880 NEXT I
890 REM G MATRİSİNİN NORMUNA 1 EKLENMESİ
900 TOP=TOP+1
910 REM G MATRİSİNİN NORM + 1'E BÖLÜNEREK H MATRİSİNİN
OLUŞTURULMASI
920 FOR I=1 TO M-N
930 FOR J=1 TO N
940 H(I,J)=G(I,J)/TOP
950 NEXT J
960 NEXT I
970 REM H MATRİSİNİN TRANSPOZESİ P MATRİSİNİN
OLUŞTURULMASI
980 FOR I=1 TO M-N
990 FOR J=1 TO N
1000 P(J,I)=H(I,J)
1010 NEXT J
1020 NEXT I
1030 REM P VE C'NIN ÇARPIM MATRİSİ Q'NUN OLUŞTURULMASI
1040 FOR I=1 TO N
1050 FOR J=1 TO N
1060 FOR K=1 TO M-N: Q(I,J)= Q(I,J)+P(I,K)*C(K,J): NEXT K
1070 NEXT J
1080 NEXT I
1090 REM N X N'LIK BİRİM MATRİS OLUŞTURMA
1100 FOR I=1 TO N
1110 FOR J=1 TO N
1120 IF I=J THEN R(I,J)=1 ELSE R(I,J)=0
1130 NEXT J
1140 NEXT I
1150 REM R - Q'NUN S DE OLUŞTURULMASI
1160 FOR I=1 TO N
1170 FOR J=1 TO N
```

```
1180 S(I,J)=R(I,J) - Q(I,J)
1190 NEXT J
1200 NEXT I
1210 REM S'NIN A MATRİSİNİN EN BÜYÜK KARE MATRİSİ B'NİN
TERSİYLE ÇARPILARAK T'NİN OLUŞTURULMASI
1220 FOR I=1 TO N
1230 FOR J=1 TO N
1240 FOR K=1 TO N: T(I,J)=T(I,J)+ S(I,K)*D(K,J): NEXT K
1250 NEXT J
1260 NEXT I
1270 REM HESAPLANAN T VE P YERLEŞTİRİLMESİYLE Z TERS
MATRİSİNİN OLUŞTURULMASI
1280 REM 1) T MATRİSİNİN Z'YE AKTARILMASI
1290 FOR I=1 TO N
1300 FOR J=1 TO N
1310 Z(I,J)=T(I,J)
1320 NEXT J
1330 NEXT I
1340 REM 2) Z'DEKİ T MATRİSİNİN SAĞINA P MATRİSİNİN
YERLEŞTİRİLMESİ
1350 FOR I=1 TO N
1360 FOR J=1 TO M-N
1370 Z(I,N+J)=P(I,J)
1380 NEXT J
1390 NEXT I
1400 PRINT: PRINT: PRINT "TERS MATRİS": PRINT "................":PRINT
1410 REM (mxn)'LIK A'NIN TERS MATRİSİ (nxm)'LIK Z'NIN EKRANDA
GÖRÜNTÜLENMESİ
1420 FOR I=1 TO N
1430 FOR J=1 TO M
1440 PRINT USING "###.######"; Z(I,J);
1450 NEXT J
1460 PRINT
1470 NEXT I
1480 DATA 5,3
1490 DATA  1,  0, -1
1500 DATA -1,  2,  0
1510 DATA  0, -1,  1
1520 DATA  0, -2,  1
1530 DATA  1,  2,  1
1550 END
```

Programın çalıştırılmasına ilişkin örnek çıktı izleyen kesimde verilmiştir.

Verilen Matris

1.000000	1.000000	-1.000000
-1.000000	2.000000	0.000000
0.000000	-1.000000	1.000000
0.000000	-2.000000	1.000000
1.000000	2.000000	1.000000

Ters Matris

1.902206	0.921324	1.885294	-0.002206	0.019110
0.943382	0.954412	0.933824	-0.001471	0.011029
0.921324	0.936765	1.907353	-0.001471	0.015441

YARARLANILAN KAYNAKLAR

HACISALİHOĞLU H. H., **Lineer Cebir**, (Gazi Üniversitesi Fen-Edebiyat Fakultesi Yayınları: Ankara), 1980.

AYRES F., (Çev.: ORAL G.), **Teori ve Problemlerle Matrisler**, (Güven Kitapevi: Ankara), 1980.

HERSTEIN I. N., **Matrix Theory and Linear Algebra**, (MacMillan Pub. Co.: New York), 1988.

FADDEEV D. K., (Çev: WILLIAMS R. C.), **Computational Methods of Linear Algebra**, (W. H. Freeman: San Fransisco), 1963.

FINKBEINER D. T., **Introduction to Matrices and Linear Transformation**, (W. H. Freeman: San Francisco), 1966.

SMITH L., (Çev: GÖĞÜŞ M., vd...), **Lineer Cebir**, (Anadolu Üniversitesi ESBAV Yayını: Eskişehir), 1993.

HÜSEYİN Ö., SEZER E., **Elementary Matrix Algebra and Differential Equations**, (Metu: Ankara), 1980.